Technische Universität München
Zentrum Mathematik
Klaus-Peter Schultze

Relative Häufigkeit und Wahrscheinlichkeit

Mit dem File **relHäufigkeit_Wahrscheinlichkeit.ggb** soll ausgehend von den relativen Häufigkeiten der Ereignisse "Augenzahl 1", "Augenzahl 2" usw. beim mehrmaligen Werfen eines Laplace-Würfels zum Wahrscheinlichkeitsbegriff hingeführt werden.

Die Simulation eines einmaligen Würfelwurfs erfolgt mit dem Befehl *Zufallszahl[1, 6]*.
Nun werde der virtuelle Würfel 6-mal geworfen. Dabei ergibt sich eine zufällige Folge f1 von 6 Augenzahlen. Wir erzeugen diese Folge mit
f1=Folge[Zufallszahl[1, 6], k, 1, 6] und eine weitere Folge f2 mit 30 Augenzahlen:
f2=Folge[Zufallszahl[1, 6], k, 1, 30]

Mit dem Befehl *Tabellentext[{f1,f2},"v"]* werden die beiden Zufallsfolgen im Grafikfenster vertikal ausgegeben.

Um eine neue Simulation des Würfelwerfens zu generieren, erzeugen wir eine Schaltfläche mit dem Namen "Neue Simulation" und tragen in deren Skripteditor (Eigenschaften -> Skripting) "bei Mausklick" den Befehl *AktualisiereKonstruktion[]* ein.

Letztlich soll bei jeder Simulation insgesamt 7776-mal bzw. 6^5-mal gewürfelt werden, wobei die ersten 6 Augenzahlen in der Folge v1 stehen sollen, die ersten 6^2 Augenzahlen (inklusive der ersten 6 Augenzahlen) in der Folge v2 usw. bis zur Folge v5, die alle 7776 bzw. 6^5 Augenzahlen enthalten soll. Dazu werden der Reihe nach die Folgen f3, f4, f5 und v1 bis v5 wie folgt erzeugt:
f3=Folge[Zufallszahl[1, 6], k, 1, 180] *f4=Folge[Zufallszahl[1, 6], k, 1, 1080]*
f5=Folge[Zufallszahl[1, 6], k, 1, 6480]
v1=f1
v2=Verbinde[{v1, f2}] *v3=Verbinde[{v2, f3}]*
v4=Verbinde[{v3, f4}] *v5=Verbinde[{v4, f5}]*

Mit dem Befehl *ZähleWenn[x == 4, v2]* erhält man die Anzahl der Würfe mit "Augenzahl 4" unter den ersten 6^2 Würfen.
Um eine Liste h2 zu bekommen, in der der Reihe nach die absoluten Häufigkeiten für die Ereignisse "Augenzahl 1", "Augenzahl 2", ... , "Augenzahl 6" unter den ersten 6^2 Würfen stehen, schreiben wir:
h2=Folge[ZähleWenn[x == i, v2], i, 1, 6]

Entsprechend schreibt man:
h1=Folge[ZähleWenn[x == i, v1], i, 1, 6]
h3=Folge[ZähleWenn[x == i, v3], i, 1, 6]
h4=Folge[ZähleWenn[x == i, v4], i, 1, 6]
h5=Folge[ZähleWenn[x == i, v5], i, 1, 6]

Die entsprechenden Listen der relativen Häufigkeiten erhält man mit:
rh1=h1/6 *rh2=h2/36* *rh3=h3/216* *rh4=h4/1296* *rh5=h5/7776*

Wie fassen die absoluten Häufigkeiten in der Matrix "hMatrix" und die relativen Häufigkeiten in der Matrix "rhMatrix" zusammen:
hMatrix={h1,h2,h3,h4,h5} *rhMatrix={rh1,rh2,rh3,rh4,rh5}*

Zur Bezeichnung der Spalten und Zeilen der Matrizen dienen folgende Listen:
Augenzahlen={{"eins", "zwei", "drei", "vier", "fünf", "sechs"}} Achtung: Doppelklammer !!
WürfeZahl = {{"", "6 x", "36 x", "216 x", "1296 x", "7776 x"}}

Mit dem Befehl "Tabellentext" lassen sich die Matrizen (die mit den Listen Augenzahlen und WürfeZahl ergänzt werden) im Grafikfenster ausgeben:

TabellenText[Transponiere[Verbinde[WürfeZahl,Transponiere[Verbinde[Augenzahlen, hMatrix]]]]]
TabellenText[Transponiere[Verbinde[WürfeZahl,Transponiere[Verbinde[Augenzahlen, rhMatrix]]]]]

	eins	zwei	drei	vier	fünf	sechs
6 x	3	0	1	1	0	1
36 x	9	5	10	5	4	3
216 x	46	32	44	31	26	37
1296 x	217	204	214	221	211	229
7776 x	1275	1286	1304	1312	1323	1276

	eins	zwei	drei	vier	fünf	sechs
6 x	0.5	0	0.167	0.167	0	0.167
36 x	0.25	0.139	0.278	0.139	0.111	0.083
216 x	0.213	0.148	0.204	0.144	0.12	0.171
1296 x	0.167	0.157	0.165	0.171	0.163	0.177
7776 x	0.164	0.165	0.168	0.169	0.17	0.164

Mit dem Befehl *Q1=Folge[(i - 1, Element[rhMatrix, i, 1]), i, 1, 5]*
wird eine Liste Q1 von 5 Punkten mit den x-Werten 0, 1, ... , 4 erstellt, deren y-Werte die relativen Häufigkeiten sind von "Augenzahl 1" nach 6^1, 6^2, ... , 6^5 Würfen.
Mit dem Befehl *s1= Streckenzug[Q1]*
werden die 5 Punkte aus der Liste Q1 mit einem Streckenzug verbunden, der die Entwicklung der relativen Häufigkeiten visualisiert.



Entsprechend definieren wir:
Q2=Folge[(i - 1, Element[rhMatrix, i, 2]), i, 1, 5] , *s2= Streckenzug[Q2]*
Q3=Folge[(i - 1, Element[rhMatrix, i, 3]), i, 1, 5] , *s3= Streckenzug[Q3]*
Q4=Folge[(i - 1, Element[rhMatrix, i, 4]), i, 1, 5] , *s4= Streckenzug[Q4]*
Q5=Folge[(i - 1, Element[rhMatrix, i, 5]), i, 1, 5] , *s5= Streckenzug[Q5]*
Q6=Folge[(i - 1, Element[rhMatrix, i, 6]), i, 1, 5] , *s6= Streckenzug[Q6]*



Letztlich erkennt man empirisch, dass sich die relativen Häufigkeiten der Ereignisse "Augenzahl 1", "Augenzahl 2", usw. mit wachsender Zahl der Würfe einer bestimmten Zahl annähern, die zumindest intuitiv als Wahrscheinlichkeit für die Ereignisse gedeutet werden kann. Leider kann die Wahrscheinlichkeit aber nicht als Grenzwert der relativen Häufigkeiten für $n \to \infty$ definiert werden, da sich die Konvergenz von relativen Häufigkeiten nicht beweisen lässt.

Technische Universität München
Zentrum Mathematik
Klaus-Peter Schultze

Vierfeldertafel - Baumdiagramm – Unabhängigkeit von Ereignissen

Wir betrachten eine Menge M von Studenten mit den Merkmalen "Blond" (B) bzw. "Nicht-Blond" ($\overline{B}$) und "Kurzsichtig" (K) bzw. "Nicht-Kurzsichtig" ($\overline{K}$).

a, b, c und d seien die Mächtigkeiten der Teilmengen $B \cap K$, $\overline{B} \cap K$, $B \cap \overline{K}$ und $\overline{B} \cap \overline{K}$.
In der Eingabezeile des Files „Vierfeldertafel etc.ggb“ deklarieren wir die entsprechenden Variablen mit *a=42*, *b=28*, *c=18* und *d=22* (insgesamt enthält M somit 110 Studenten), die durch jeweiliges Klicken auf den Radiobutton als Schieberegler erscheinen. Unter „Eigenschaften -> Schieberegler" begrenzen wir die Schieberegler jeweils auf das Intervall von min=0 bis max=50 und legen als Schrittweite jeweils den Wert 1 fest.

Wird nun ein/e Student/in aus der Menge M ausgelost, so gilt für die Wahrscheinlichkeit, dass er/sie blond ist: P(B) = (a+c)/(a+b+c+d).

Diese Wahrscheinlichkeit bezeichnen wir mit p_1 und definieren deshalb:

P(B): *p_1 = (a+c)/(a+b+c+d)* Entsprechend werden weitere Wahrscheinlichkeiten definiert:

$P(K\,|\,B)$ = P("K unter der Bedingung B“): *p_2 = a/(a+c)* $P(K\,|\,\overline{B})$: *p_3 = b/(b+d)*

P(K): *pu_1 = (a+b)/(a+b+c+d)* $P(B\,|\,K)$: *pu_2 = a/(a+b)* $P(B\,|\,\overline{K})$: *pu_3 = c/(c+d)*

Die Bezeichnungen der Merkmale wird in den Textvariablen T1 und T2 festgelegt mit:
T1="B" bzw. *T2="K"*.

Mit Hilfe der **Makros Vierfeldertafel**, **Baumkomplett**, **Wahrscheinlichkeitstafel1** und **Wahrscheinlichkeitstafel2** werden insgesamt fünf dynamisch-grafische Objekte erzeugt.

Dazu werden im Grafikfenster noch fünf bewegliche Punkte A, B, C, D und E benötigt:
A (im oberen linken Bereich), B (im oberen mittleren Bereich), C (im oberen rechten Bereich), D (im unteren linken Bereich) und E (im unteren rechten Bereich).

Für die Einstellung der Größen der Objekte muss noch ein Parameter k mit z.B. *k=0.9* deklariert werden. Durch Veränderung von k (eventuell mit einem Schieberegler) lassen sich die Objekte nach ihrer Erzeugung je nach Wunsch noch größer oder kleiner machen.



Die fünf Objekte können mit den folgenden Befehlen oder mit Hilfe der Makrobuttons erzeugt werden:

Wahrscheinlichkeitstafel1[A, xAchse, k, p_1, p_2, p_3, T1, T2]
Vierfeldertafel[B, xAchse, k, a, b, c, d, T1, T2]
Wahrscheinlichkeitstafel2[C, xAchse, k, pu_1, pu_2, pu_3, T1, T2]
Baumkomplett[D, xAchse, k, p_1, p_2, p_3, T1, T2]
Baumkomplett[E, xAchse, k, pu_1, pu_2, pu_3, T2, T1]

Durch Verschieben der „Leitpunkte“ A, B, C, D und E und Einstellung der Größen mit k lassen sich die fünf Objekte wunschgemäß anordnen. Auch die Bezeichnungen für die zugrunde liegenden Merkmale bzw. Ereignisse können mit den Textvariablen T1 und T2 je nach Anwendung bei konkreten Aufgaben frei gewählt werden. Letztlich sollten die Leitpunkte und k im Grafikfenster aber unsichtbar gemacht werden, um ein versehentliches Verstellen der gewählten Anordnung zu vermeiden.

Die 5 Makros sind im File **Vierfeldertafel-etc.ggb** bereitgestellt. Die zugehörigen Files stehen im Ordner Makros sowohl als ggb- als auch als ggt-File zur Verfügung. Um diese Makros in einem neuen GeoGebra-Fenster verfügbar zu machen, muss man die ggt-Files mit „drag and drop“ ins Zeichenfenster setzen.

Die Wahrscheinlichkeitstafeln sind Quadrate mit der Seitenlänge 1, die jeweils in vier disjunkte Rechtecke zerlegt sind, die den Ereignissen $B \cap K$, $\overline{B} \cap K$, $B \cap \overline{K}$ und $\overline{B} \cap \overline{K}$ entsprechen. Die (Maßzahlen der) Flächeninhalte der Rechtecke sind gleich den Wahrscheinlichkeiten der entsprechenden Ereignisse. Auch die Seitenlängen der Rechtecke sind als Wahrscheinlichkeiten zu interpretieren. So sind die Seitenlängen des zum Ereignis $B \cap K$ gehörenden Rechtecks in der oberen linken Tafel gleich den Wahrscheinlichkeiten $P(B)$ und $P(K \mid B)$. Aus der Flächenformel für das Rechteck folgt daraus:

$P(B \cap K) = P(B) \cdot P(K \mid B)$ Das ist eine Pfadregel im Baum unten links.

Entsprechend folgert man für das gleiche Ereignis $B \cap K$ bzw. $K \cap B$ aus der Tafel oben rechts: $P(K \cap B) = P(K) \cdot P(B \mid K)$ Das ist eine Pfadregel im Baum unten rechts.

Vermindert man mit dem Schieberegler den Wert von $d = |\overline{B} \cap \overline{K}|$ von 22 auf 12, so ist die Zerlegung in Rechtecke in den beiden Tafeln genau gleichgestaltig. Aus der linken Tafel erkennt man: $P(K \mid B) = P(K \mid \overline{B})$

und zusammen mit der rechten Tafel: $P(K \mid B) = P(K \mid \overline{B}) = P(K)$

Diese Gleichheit lässt sich auch mit den beiden Bäumen entdecken.
Damit ist der Anteil der Kurzsichtigen unter den Blonden genauso groß, wie unter den Nicht-Blonden, d.h.: Die Ereignisse B und K sind unabhängig.
Aus der allgemeingültigen Pfadregel $P(B \cap K) = P(B) \cdot P(K \mid B)$
wird unter der Bedingung $P(K \mid B) = P(K)$, die nur bei Unabhängigkeit von B und K gilt,
die übliche Unabhängigkeitsbedingung: $\boxed{P(B \cap K) = P(B) \cdot P(K)}$ für zwei Ereignisse B und K.

Letztlich erkennt man, dass, unter der Voraussetzung der Unabhängigkeit, in beiden Tafeln die vier Rechtecke nur von Seiten mit den vier Längen (Wahrscheinlichkeiten) $P(B)$, $P(\overline{B})$, $P(K)$ und $P(\overline{K})$ eingeschlossen werden: Daraus folgt mit der Flächenformel unmittelbar:

$P(B \cap K) = P(B) \cdot P(K)$; $P(\overline{B} \cap K) = P(\overline{B}) \cdot P(K)$;

$P(B \cap \overline{K}) = P(B) \cdot P(\overline{K})$ und $P(\overline{B} \cap \overline{K}) = P(\overline{B}) \cdot P(\overline{K})$.

Es zeigt sich, dass mit der Unabhängigkeit des Ereignispaares (B,K) auch die Ereignispaare $(B, \overline{K})$, $(\overline{B}, K)$ und $(\overline{B}, \overline{K})$ unabhängig sein müssen.

Durch Veränderung der Parameter a, b, c und d, mittels der Schieberegler und Beobachtung der Tafeln und Bäume, lassen sich noch viele weitere mathematische Regeln entdecken bzw. bestätigen. Das Verständnis für die vielfältigen Zusammenhänge wird besonders dadurch gefördert, dass jeweils fünf unterschiedliche Darstellungen der gleichen mathematischen Struktur zur Verfügung stehen.

Technische Universität München
Zentrum Mathematik
Klaus-Peter Schultze/Hermann Vogel

Bernoulliketten – Binomialverteilung - Normalverteilung

Im File **Binomial-Normalverteilung.ggb** deklarieren wir zunächst die Parameter *n = 30* und *p = 0.4*, die durch jeweiliges Klicken auf den Radiobutton als Schieberegler erscheinen. Unter "Eigenschaften -> Schieberegler" begrenzen wir den Schieberegler von n auf das Intervall von min=1 bis max=50 und den Schieberegler von p von min=0 bis max=1. Als Schrittweite legen wir bei n den Wert 1 und bei p den Wert 0.01 fest.
Mit dem Befehl *b = Binomial[n, p]* wird das **Standard-Histogramm** der Wahrscheinlichkeits-verteilung der Bernoullikette mit der Länge n und der Trefferwahrscheinlichkeit p erzeugt.
Das Histogramm ist nach oben begrenzt durch den Graphen einer nicht-stetigen, "treppen-förmigen" Funktion (der sog. Dichtefunktion des Histogramms), die binom heißen soll und für alle $x \in [-0.5 \; ; \; n+0.5]$ definiert ist. Der Funktionsterm entspricht der bekannten Bernoulliformel: $B(n,p,k) = \binom{n}{k} \cdot p^k \cdot (1-p)^{n-k}$ mit k=round(x).

Für den Binomialkoeffizienten "k aus n" gilt: $\binom{n}{k} = \frac{n!}{k! \cdot (n-k)!}$.

Entsprechend definieren wir folgende Funktionen:

Fakultät: *fakult(x) = round(x) !*
Binomialkoeffizient: *binomko(x) = fakult(n) / fakult(x) / fakult(n - x)*
und schließlich: *binom(x) = binomko(x)*p^round(x)*(1 - p)^(n - round(x))*

Den zugehörigen Erwartungswert μ und die Standardabweichung σ erhalten wir mit *μ=n*p* und *σ=sqrt(n*p*(1-p))*
Beim sog. Standardisieren der Dichtefunktion binom erhält man eine verwandte Funktion, wir nennen sie binomstand, deren Erwartungswert 0 und deren Standardabweichung 1 ist. Eine bestimmte Fläche unter dem Graphen der Funktion binom hat den gleichen Inhalt wie die entsprechende Fläche unter dem Graphen von binomstand, da beim Standardisieren der Graph zunächst um μ nach links verschoben wird und anschließend zugleich eine Stau-chung/Streckung des Graphen in x-Richtung (Streckungsfaktor: 1/ σ) und eine Streckung/ Stauchung in y-Richtung (Streckungsfaktor: σ) erfolgt.

Wie definieren deshalb: *binomstand(x) = σ*binom(x*σ + μ)*

Erhöht man nun die Länge n der Bernoullikette, so nähern sich (für alle $p \in]0;1[$) die standardisierten Dichtefunktionen der gleichen Grenzfunktion φ an, die als Gaußsche Funktion oder auch Gaußsche Glockenkurve bezeichnet wird.

Für φ gilt: *φ(x) = 1 / sqrt(2π)*exp(-0.5*x²)* , vgl. **Binomial-Normalverteilung-Lsg1.ggb.**



Ist die standardisierte Dichtefunktion einer Zufallsgröße gleich φ, so heißt sie normalverteilt. Die binomialverteilten Zufallsgrößen (z.B. Bernoulliketten) sind für größere Werte von n (Faustregel: np(1-p) > 9) näherungsweise normalverteilt. D.h. die "Bernoulliwahrscheinlichkeiten" B(n;p; $k_1 \leq k \leq k_2$) lassen sich dann näherungsweise durch Integration von φ berechnen.

Die Genauigkeit dieser näherungsweisen Berechnung soll experimentell untersucht werden. Dazu deklarieren wir zunächst die Parameter *k_1=9* und *k_2=13* und klicken jeweils auf den zugehörigen Radiobutton, um Schieberegler zu generieren. Unter "Eigenschaften -> Schieberegler" begrenzen wir den Schieberegler von k_1 auf das Intervall von min=0 bis max=n und den Schieberegler von k_2 von min=k_1 bis max=n. Als Schrittweite legen wir jeweils 1 fest.

Anschließend berechnen wir folgende Integrale bzw. Wahrscheinlichkeiten:
W1=Integral[binom, k_1 - 0.5, k_2 + 0.5]
W2=Integral[binomstand, (k_1 - 0.5 - μ) / σ, (k_2 + 0.5 - μ) / σ]
W3=Integral[φ, (k_1 - 0.5 - μ) / σ, (k_2 + 0.5 - μ) / σ]
Wegen der Flächengleichheit müssen W1 und W2 gleich sein. W3 ist der Näherungswert mittels φ, vgl. **Binomial-Normalverteilung-Lsg2.ggb**.



GeoGebra kann die Binomialkoeffizienten (noch) nicht als Funktion von x verarbeiten. Deswegen wurden diese über die Fakultätsfunktion definiert. Im CAS-Teil geht dies aber schon.

Die Bernoulliformel wird im CAGD (Computer Aided Geometric Design) zur Definition von Freiformkurven verwendet, vgl. **BezierKurve.ggb**.